```
            pll=(x*(2*ll-1)*pmmp1-(ll+m-1)*pmm)/(ll-m)
            pmm=pmmp1
            pmmp1=pll
        enddo 12
        plgndr=pll
    endif
endif
return
END
```

CITED REFERENCES AND FURTHER READING:

Magnus, W., and Oberhettinger, F. 1949, *Formulas and Theorems for the Functions of Mathematical Physics* (New York: Chelsea), pp. 54ff. [1]

Abramowitz, M., and Stegun, I.A. 1964, *Handbook of Mathematical Functions*, Applied Mathematics Series, Volume 55 (Washington: National Bureau of Standards; reprinted 1968 by Dover Publications, New York), Chapter 8. [2]

6.9 Fresnel Integrals, Cosine and Sine Integrals

Fresnel Integrals

The two Fresnel integrals are defined by

$$C(x) = \int_0^x \cos\left(\frac{\pi}{2}t^2\right) dt, \qquad S(x) = \int_0^x \sin\left(\frac{\pi}{2}t^2\right) dt \tag{6.9.1}$$

The most convenient way of evaluating these functions to arbitrary precision is to use power series for small x and a continued fraction for large x. The series are

$$\begin{aligned} C(x) &= x - \left(\frac{\pi}{2}\right)^2 \frac{x^5}{5\cdot 2!} + \left(\frac{\pi}{2}\right)^4 \frac{x^9}{9\cdot 4!} - \cdots \\ S(x) &= \left(\frac{\pi}{2}\right) \frac{x^3}{3\cdot 1!} - \left(\frac{\pi}{2}\right)^3 \frac{x^7}{7\cdot 3!} + \left(\frac{\pi}{2}\right)^5 \frac{x^{11}}{11\cdot 5!} - \cdots \end{aligned} \tag{6.9.2}$$

There is a complex continued fraction that yields both $S(x)$ and $C(x)$ simultaneously:

$$C(x) + iS(x) = \frac{1+i}{2}\,\mathrm{erf}\, z, \qquad z = \frac{\sqrt{\pi}}{2}(1-i)x \tag{6.9.3}$$

where

$$\begin{aligned} e^{z^2}\,\mathrm{erfc}\, z &= \frac{1}{\sqrt{\pi}}\left(\frac{1}{z+}\,\frac{1/2}{z+}\,\frac{1}{z+}\,\frac{3/2}{z+}\,\frac{2}{z+}\cdots\right) \\ &= \frac{2z}{\sqrt{\pi}}\left(\frac{1}{2z^2+1-}\,\frac{1\cdot 2}{2z^2+5-}\,\frac{3\cdot 4}{2z^2+9-}\cdots\right) \end{aligned} \tag{6.9.4}$$

In the last line we have converted the "standard" form of the continued fraction to its "even" form (see §5.2), which converges twice as fast. We must be careful not to evaluate the alternating series (6.9.2) at too large a value of x; inspection of the terms shows that $x = 1.5$ is a good point to switch over to the continued fraction.

Note that for large x

$$C(x) \sim \frac{1}{2} + \frac{1}{\pi x}\sin\left(\frac{\pi}{2}x^2\right), \qquad S(x) \sim \frac{1}{2} - \frac{1}{\pi x}\cos\left(\frac{\pi}{2}x^2\right) \tag{6.9.5}$$

Thus the precision of the routine `frenel` may be limited by the precision of the library routines for sine and cosine for large x.

```
      SUBROUTINE frenel(x,s,c)
      INTEGER MAXIT
      REAL c,s,x,EPS,FPMIN,PI,PIBY2,XMIN
      PARAMETER (EPS=6.e-8,MAXIT=100,FPMIN=1.e-30,XMIN=1.5,
     *     PI=3.1415927,PIBY2=1.5707963)
```

Computes the Fresnel integrals $S(x)$ and $C(x)$ for all real x.
Parameters: EPS is the relative error; MAXIT is the maximum number of iterations allowed; FPMIN is a number near the smallest representable floating-point number; XMIN is the dividing line between using the series and continued fraction; $\mathrm{PI} = \pi$; $\mathrm{PIBY2} = \pi/2$.

```
      INTEGER k,n
      REAL a,absc,ax,fact,pix2,sign,sum,sumc,sums,term,test
      COMPLEX b,cc,d,h,del,cs
      LOGICAL odd
      absc(h)=abs(real(h))+abs(aimag(h))     Statement function.
      ax=abs(x)
      if(ax.lt.sqrt(FPMIN))then              Special case: avoid failure of convergence test
        s=0.                                   because of underflow.
        c=ax
      else if(ax.le.XMIN)then                Evaluate both series simultaneously.
        sum=0.
        sums=0.
        sumc=ax
        sign=1.
        fact=PIBY2*ax*ax
        odd=.true.
        term=ax
        n=3
        do 11 k=1,MAXIT
          term=term*fact/k
          sum=sum+sign*term/n
          test=abs(sum)*EPS
          if(odd)then
            sign=-sign
            sums=sum
            sum=sumc
          else
            sumc=sum
            sum=sums
          endif
          if(term.lt.test)goto 1
          odd=.not.odd
          n=n+2
        enddo 11
        pause 'series failed in frenel'
1       s=sums
        c=sumc
      else                                   Evaluate continued fraction by modified Lentz's
        pix2=PI*ax*ax                          method (§5.2).
        b=cmplx(1.,-pix2)
```

```
    cc=1./FPMIN
    d=1./b
    h=d
    n=-1
    do 12 k=2,MAXIT
        n=n+2
        a=-n*(n+1)
        b=b+4.
        d=1./(a*d+b)                    Denominators cannot be zero.
        cc=b+a/cc
        del=cc*d
        h=h*del
        if(absc(del-1.).lt.EPS)goto 2
    enddo 12
    pause 'cf failed in frenel'
2   h=h*cmplx(ax,-ax)
    cs=cmplx(.5,.5)*(1.-cmplx(cos(.5*pix2),sin(.5*pix2))*h)
    c=real(cs)
    s=aimag(cs)
endif
if(x.lt.0.)then                         Use antisymmetry.
    c=-c
    s=-s
endif
return
END
```

Cosine and Sine Integrals

The cosine and sine integrals are defined by

$$\begin{aligned}\mathrm{Ci}(x) &= \gamma + \ln x + \int_0^x \frac{\cos t - 1}{t}\,dt \\ \mathrm{Si}(x) &= \int_0^x \frac{\sin t}{t}\,dt\end{aligned} \tag{6.9.6}$$

Here $\gamma \approx 0.5772\ldots$ is Euler's constant. We only need a way to calculate the functions for $x > 0$, because

$$\mathrm{Si}(-x) = -\,\mathrm{Si}(x), \qquad \mathrm{Ci}(-x) = \mathrm{Ci}(x) - i\pi \tag{6.9.7}$$

Once again we can evaluate these functions by a judicious combination of power series and complex continued fraction. The series are

$$\begin{aligned}\mathrm{Si}(x) &= x - \frac{x^3}{3\cdot 3!} + \frac{x^5}{5\cdot 5!} - \cdots \\ \mathrm{Ci}(x) &= \gamma + \ln x + \left(-\frac{x^2}{2\cdot 2!} + \frac{x^4}{4\cdot 4!} - \cdots\right)\end{aligned} \tag{6.9.8}$$

The continued fraction for the exponential integral $E_1(ix)$ is

$$\begin{aligned}E_1(ix) &= -\,\mathrm{Ci}(x) + i[\mathrm{Si}(x) - \pi/2] \\ &= e^{-ix}\left(\frac{1}{ix+}\;\frac{1}{1+}\;\frac{1}{ix+}\;\frac{2}{1+}\;\frac{2}{ix+}\cdots\right) \\ &= e^{-ix}\left(\frac{1}{1+ix-}\;\frac{1^2}{3+ix-}\;\frac{2^2}{5+ix-}\cdots\right)\end{aligned} \tag{6.9.9}$$

The "even" form of the continued fraction is given in the last line and converges twice as fast for about the same amount of computation. A good crossover point from the alternating series to the continued fraction is $x = 2$ in this case. As for the Fresnel integrals, for large x the precision may be limited by the precision of the sine and cosine routines.

```
      SUBROUTINE cisi(x,ci,si)
      INTEGER MAXIT
      REAL ci,si,x,EPS,EULER,PIBY2,FPMIN,TMIN
      PARAMETER (EPS=6.e-8,EULER=.57721566,MAXIT=100,PIBY2=1.5707963,
     *     FPMIN=1.e-30,TMIN=2.)
C     Computes the cosine and sine integrals Ci(x) and Si(x). Ci(0) is returned as a large negative
C     number and no error message is generated. For x < 0 the routine returns Ci(-x) and you
C     must supply the -iπ yourself.
C     Parameters: EPS is the relative error, or absolute error near a zero of Ci(x); EULER = γ;
C     MAXIT is the maximum number of iterations allowed; PIBY2 = π/2; FPMIN is a number
C     near the smallest representable floating-point number; TMIN is the dividing line between
C     using the series and continued fraction.
      INTEGER i,k
      REAL a,err,fact,sign,sum,sumc,sums,t,term,absc
      COMPLEX h,b,c,d,del
      LOGICAL odd
      absc(h)=abs(real(h))+abs(aimag(h))      Statement function.
      t=abs(x)
      if(t.eq.0.)then                         Special case.
        si=0.
        ci=-1./FPMIN
        return
      endif
      if(t.gt.TMIN)then                       Evaluate continued fraction by modified Lentz's
        b=cmplx(1.,t)                            method (§5.2).
        c=1./FPMIN
        d=1./b
        h=d
        do 11 i=2,MAXIT
          a=-(i-1)**2
          b=b+2.
          d=1./(a*d+b)                        Denominators cannot be zero.
          c=b+a/c
          del=c*d
          h=h*del
          if(absc(del-1.).lt.EPS)goto 1
        enddo 11
        pause 'cf failed in cisi'
1       continue
        h=cmplx(cos(t),-sin(t))*h
        ci=-real(h)
        si=PIBY2+aimag(h)
      else                                    Evaluate both series simultaneously.
        if(t.lt.sqrt(FPMIN))then              Special case: avoid failure of convergence test
          sumc=0.                                because of underflow.
          sums=t
        else
          sum=0.
          sums=0.
          sumc=0.
          sign=1.
          fact=1.
          odd=.true.
          do 12 k=1,MAXIT
            fact=fact*t/k
            term=fact/k
```

```
                sum=sum+sign*term
                err=term/abs(sum)
                if(odd)then
                    sign=-sign
                    sums=sum
                    sum=sumc
                else
                    sumc=sum
                    sum=sums
                endif
                if(err.lt.EPS)goto 2
                odd=.not.odd
            enddo 12
            pause 'maxits exceeded in cisi'
        endif
2       si=sums
        ci=sumc+log(t)+EULER
    endif
    if(x.lt.0.)si=-si
    return
    END
```

CITED REFERENCES AND FURTHER READING:

Stegun, I.A., and Zucker, R. 1976, *Journal of Research of the National Bureau of Standards*, vol. 80B, pp. 291–311; 1981, *op. cit.*, vol. 86, pp. 661–686.

Abramowitz, M., and Stegun, I.A. 1964, *Handbook of Mathematical Functions*, Applied Mathematics Series, Volume 55 (Washington: National Bureau of Standards; reprinted 1968 by Dover Publications, New York), Chapters 5 and 7.

6.10 Dawson's Integral

Dawson's Integral $F(x)$ is defined by

$$F(x) = e^{-x^2} \int_0^x e^{t^2}\, dt \tag{6.10.1}$$

The function can also be related to the complex error function by

$$F(z) = \frac{i\sqrt{\pi}}{2} e^{-z^2} \left[1 - \operatorname{erfc}(-iz)\right]. \tag{6.10.2}$$

A remarkable approximation for $F(x)$, due to Rybicki [1], is

$$F(z) = \lim_{h\to 0} \frac{1}{\sqrt{\pi}} \sum_{n \text{ odd}} \frac{e^{-(z-nh)^2}}{n} \tag{6.10.3}$$

What makes equation (6.10.3) unusual is that its accuracy increases *exponentially* as h gets small, so that quite moderate values of h (and correspondingly quite rapid convergence of the series) give very accurate approximations.